Using Mplus To Do DSEM With Cycles

Bengt Muthén
Professor Emeritus, UCLA
Mplus: https://www.statmodel.com
bmuthen@statmodel.com

Tihomir Asparouhov
Mplus

Mplus Web Talks: No. 7
May 2024

We thank Thuy Nguyen and Noah Hastings for expert assistance.

Background

- Muthén, Asparouhov & Keijsers (2024). Dynamic structural equation modeling with cycles. Submitted February, 2024. Available in Mplus Version 8.11, May 2024
  Supplemental Material shows Mplus inputs.
- Example:
  - Modeling of daily cycles and weekday effects for positive affect and tiredness among adolescents ages 12 to 16
  - Intensive longitudinal data collected by experience sampling methods
  - Several measures per day at random timepoints for seven days
  - Mplus TINTERVAL option using 3-hour bins
  - N = 240, T = 56
  - Six 7-category PA items: relaxed, satisfied, confident, happy, energetic, and excited.
  - Analysis of total PA score as well as factors behind the PA items
- Introduction to DSEM: Mplus Web Talk No. 6. Using Mplus To Do Dynamic Structural Equation Modeling. February 2023.

Prologue: PA (Red) and Tiredness (Blue)



- Does tiredness follow cycles due to circadian (24-hour) rhythm?
- Are such cycles also affecting PA?
- Is there a relationship between tiredness and PA after accounting for these cycles?

Prologue: Residual Relation

- Is there a relationship between tiredness and PA after accounting for cycles?



- The paper shows that using only two cycles variables can account for circadian rythm: cosinor model.

Prologue: Bivariate Two-Level RDSEM with Cycles



Analysis Steps Based on Table 1 of the Paper

- **Step 1**: Finding cycles and their duration
  - Cross-classified DSEM with unrestricted between time estimates
  - Plotting observed means and time-specific estimates
- **Step 2**: Fitting cycles by cosinor model, fixed cycles coefficients
  - Cross-classified DSEM with cycles duration based on Step 1
  - Plotting estimated curves using LOOP PLOT
  - Bivariate two-level RDSEM
- **Step 3**: Finding deviations from cycles
  - Cross-classified RDSEM to test significance of deviations
  - Adding and plotting effects of time-specific dummy variables
- **Step 4**: Fitting cycles, random cycles coefficients
  - Two-level and cross-classified RDSEM
- **Step 5**: Explaining random cycles coefficients by covariates
  - Multiple imputation, translation to amplitude and phase

Step 1: Cross-Classified DSEM



Time Series Plot of Between Time Estimates (Blue) and PA Sample Means Over Subjects (Red)



- How was it done?

Step 1a: Cross-Classified DSEM Input




```
USEVAR = pa;
CLUSTER = id time;
TINTERVAL = hrs (3 time);
LAGGED = pa(1);
ANALYSIS:   TYPE = CROSSCLASSIFIED;
ESTIMATOR = BAYES;
BITERATIONS = (2000);
PROCESSORS = 2;
MODEL:      %WITHIN%
pa ON pa&1;
%BETWEEN id%
pa;
%BETWEEN time%
pa;
PLOT:       TYPE = PLOT3;
FACTORS = ALL(200);
```

Demonstration of Step 1a Output and Plotting

- The step number refers to the analysis steps of Table 1 in the Muthén, Asparouhov, Keijsers (2024) paper and the inputs given in the Supplementary Material
- The Supplementary Material shows more inputs than discussed in Web Talk 7

Step 2: Cross-Classified DSEM with Cycles



Step 2: Basic Cosinor Model

The aim is to fit the cycles by regressing $y(t)$ at timepoint t on the two covariates $x_{1t} = \sin(2\pi\omega t)$ and $x_{2t} = \cos(2\pi\omega t)$,

$$y(t) = \beta_0 + \beta_1 x_{1t} + \beta_2 x_{2t} + \zeta_t, \tag{1}$$

where the cycles coefficients β_1 and β_2 carry information about the amplitude (A) and phase (ϕ),

$$\beta_1 = A\ \sin(2\pi\omega\phi), \tag{2}$$

$$\beta_2 = A\ \cos(2\pi\omega\phi), \tag{3}$$

where A is the amplitude defined as half the difference between the highest and lowest y values, ϕ is a phase shift, ω is a frequency index, and

$$A = \sqrt{\beta_1^2 + \beta_2^2}, \tag{4}$$

$$\phi = tan^{-1}(\beta_1/\beta_2). \tag{5}$$

- See, however, Section 2 of the Supplemental Material

Cosinor Curves ($\omega = 1/8$)





Step 2: Cosinor Model Versions in Mplus

Models and number of parameters for univariate case (run numbers refer to Table 1 of paper):

- Two-level RDSEM, standard cosinor model (cycles on within):
  - Fixed cycles coefficients (slide 12 model with AR for residual): 6 parameters (run 2c)
  - Random cycles coefficients: 11 parameters (run 4a)
- Cross-classified DSEM (cycles on between time level):
  - Fixed cycles coefficients (slide 11 model): 7 parameters (run 2a)
  - Random cycles coefficients: Not available
- Cross-classified RDSEM (cycles on within):
  - Fixed cycles coefficients: 7 parameters (run 3b)
  - Random cycles coefficients: 12 parameters (run 4c)

The models used in the cross-classified, fixed coefficient runs 2a and 3b are equivalent. The benefits of the 3b approach will be discussed when looking for significant deviations from cycles.

Step 2a: Cross-Classified DSEM Cosinor Analysis. Plot of Cycles (Red) and Between Time Estimates (Blue)



- Between Time $R^2(PA) = 0.41$

Step 2a: Cross-classified DSEM Input

```
                    USEVAR = pa x1 x2;
                    CLUSTER = id time;
                    TINTERVAL = hrs (3 time);
                    BETWEEN = (time) x1 x2;
                    LAGGED= pa(1);

DEFINE:             x1 = SIN(6.2831853*(1/8)*time);
                    x2 = COS(6.2831853*(1/8)*time);

ANALYSIS:           TYPE = CROSSCLASSIFIED;
                    ESTIMATOR = BAYES;
                    BITERATIONS = (2000);
                    THIN = 10;
                    PROCESSORS = 2;
```

Step 2a: Cross-classified DSEM Input, Continued

```
MODEL:              %WITHIN%
                    pa ON pa&1;

                    %BETWEEN id%
                    pa;
                    [pa] (p0);

                    %BETWEEN time%
                    pa ON x1 (p1)
                    x2 (p2);

MODEL CONSTRAINT:   LOOP(time,1,56,0.1);
                    PLOT(pacycles fscycles);
                    pacycles = p0 +
                    p1*SIN(6.2831853*(1/8)*time)+ p2*COS(6.2831853*(1/8)*time);
                    fscycles =
                    p1*SIN(6.2831853*(1/8)*time)+ p2*COS(6.2831853*(1/8)*time);

OUTPUT:             STANDARDIZED TECH1 TECH4 TECH8;

PLOT:               TYPE = PLOT3;
                    FACTORS = ALL(200);
```

Demonstration of 2a Plotting

- Inserting loop plot in Time series plot
- Saving plot data
  - Column 1: x-axis values for 8 timepoints per day (1-56)
  - Column 2: Estimated B2a values
  - Column 3: x-axis values for loop plot according to the chosen steps (0.1 steps)
  - Column 4: Cycles values (repeats starting at 9, 17, etc)
- Run 1a has 2 columns: 1-56, B2a

Step 2: Cosinor Analysis. Two Runs, Cycles (Red) and Step 1 (Slide 8) Between Time Estimates (Blue)



Step 3: Finding Deviations from Cycles

Three alternatives discussed in the paper:

1. Add dummy variables for certain days or timepoints and check for significance of the effects: Step 3a run with LOOP PLOT
2. BSEM approach
3. Use cross-classified RDSEM to find significant deviations from cycles: Step 3b run and saved data

- How is it done? Demo of Step 3a run

Step 3a: Cross-Classified DSEM Input

```
                USEVAR = pa x1 x2 sat;
                CLUSTER = id time;
                TINTERVAL = hrs (3 time);
                BETWEEN = (time) x1 x2 sat;
                LAGGED = pa(1);

DEFINE:         x1 = SIN(6.2831853*(1/8)*time);
                x2 = COS(6.2831853*(1/8)*time);
                IF(time>33 .AND. time<40)THEN sat = 1 ELSE sat=0;

ANALYSIS:       TYPE = CROSSCLASSIFIED;
                ESTIMATOR = BAYES;
                BITERATIONS = (2000);
                THIN = 10; PROCESSORS = 2;

MODEL:          %WITHIN%
                pa ON pa&1;
                %BETWEEN id%
                pa;
                [pa] (p0);
                %BETWEEN time%
                pa ON x1 (p1)
                x2 (p2)
                sat (p3);
```

Step 3a: Cross-Classified DSEM Input, Continued

```
MODEL CONSTRAINT:   LOOP(time,1,56,0.1);
                    PLOT(pacycles fscycles saturday patot fstot);
                    pacycles = p0 + p1*SIN(6.2831853*(1/8)*time)+
                    p2*COS(6.2831853*(1/8)*time);
                    fscycles = p1*SIN(6.2831853*(1/8)*time)+
                    p2*COS(6.2831853*(1/8)*time);
                    saturday = p3*[34,39];
                    patot = pacycles + saturday;
                    fstot = fscycles + saturday;

OUTPUT:             STANDARDIZED TECH1 TECH8;

PLOT:               TYPE = PLOT3; FACTORS = ALL(200);
```

- Demo of step 3a time series plot with Saturday effect

Step 3 (Alt 3): Finding Significant Deviations from Cycles Using Cross-Classified RDSEM (Step 3b Run)

Cycles estimated on the Within level instead of the Between Time level so that deviations can be estimated on the Between Time level:



- How is it done? Demo of Step 3b run

Step 3b: Cross-Classified RDSEM Input

```
                    USEVAR = pa x1 x2;
                    CLUSTER = id time;
                    TINTERVAL = hrs (3 time);
                    WITHIN = x1 x2;
                    LAGGED = pa(1);

DEFINE:             x1 = SIN(6.2831853*(1/8)*time);
                    x2 = COS(6.2831853*(1/8)*time);

ANALYSIS:           TYPE = CROSSCLASSIFIED;
                    ESTIMATOR = BAYES;
                    BITERATIONS = (2000);
                    PROCESSORS = 2;

MODEL:              %WITHIN%
                    pa ON x1 (p1)
                    x2 (p2);
                    pa^ ON pa^1;
                    %BETWEEN id%
                    pa;
                    [pa] (p0);
                    %BETWEEN time%
                    pa;
```

Step 3b: Cross-Classified RDSEM Input, Continued

```
MODEL CONSTRAINT:   LOOP(time,1,56,0.1);
                    PLOT(pacycles fscycles);
                    pacycles = p0 +
                    p1*SIN(6.2831853*(1/8)*time)+ p2*COS(6.2831853*(1/8)*time);
                    fscycles =
                    p1*SIN(6.2831853*(1/8)*time)+ p2*COS(6.2831853*(1/8)*time);

OUTPUT:             STANDARDIZED TECH1 TECH4 TECH8;

PLOT:               TYPE = PLOT3;
                    FACTORS = ALL(200);

SAVEDATA:           SAVE = FS(200 10);
                    FILE = fscyclesdeviations.dat;
```

Step 4: Fitting Cycles, Random Cycles Coefficients

Cosinor model with random β_1, β_2 coefficients:

- Two-level RDSEM
  - Coefficient variation on the Between level
  - Two-level analysis is suitable if small residual variance on the Between Time level is found in the cross-classified run of Step 2a
- Cross-classified RDSEM
  - Use model with cycles on the within level to allow coefficients variation on the Between ID level
  - More time consuming than two-level analysis
- Inputs are shown in Supplementary Material

Step 5: Explaining Random Cycles Coefficients by Covariates

Substantive interpretation of cycles is more clearly done by translating the β_1, β_2 cycles coefficients to amplitude and phase. This can be done by the two steps:

1. Step 5a: Two-level RDSEM with between-level covariates
   - Multiple imputation of random β_1, β_2 cycles coefficients using the SAVEDATA command. For instance, requesting 200 plausible values of the two coefficients for each person, there are 200 datasets each with N rows and the two random coefficients and the between-level covariates as columns
2. Step 5b: Single-level analysis using TYPE = IMPUTATION with translation in DEFINE of β_1, β_2 coefficients into amplitude and phase, e.g., $A_i = \sqrt{\beta_{1i}^2 + \beta_{2i}^2}$

Step 5: Cycles for Factors Behind the PA Items

Three low and three high arousal PA items:

	Between ID		Within = Between Time	
	PA Low	PA High	PA Low	PA High
Relaxed	0.94	0	0.76	0
Satisfied	1.00	0	0.86	0
Confident	0.80	0	0.73	0
Happy	0.52	0.49	0.44	0.45
Energetic	0	0.96	0	0.82
Excited	0	1.00	0	0.91

- Factor correlations are 0.83 and 0.65



- Between time R^2 values are 0.48 and 0.59 for the low- and high-arousal factors

Step 5: Two-Level RDSEM with Random Cycles Coefficients for Factors Related to Time-Invariant Covariates



- How is it done? Demo of Step 5a and Step 5b runs

Step 5a: Two-level RDSEM Factor Analysis Input

```
                USEVAR = PALA1 PALA2 PALA3 PAHA1 PAHA2 PAHA3
                age SDQ girl tiredavg x1 x2;
                CLUSTER = id;
                WITHIN = x1 x2;
                BETWEEN = girl sdq age tiredavg;
                TINTERVAL = hrs (3 time);

DEFINE:         tiredavg = CLUSTER_MEAN(tired);
                girl = sexAA - 1;
                x1 = SIN(6.2831853*(1/8)*time);
                x2 = COS(6.2831853*(1/8)*time);

                CENTER sdq age tiredavg (GRANDMEAN);

ANALYSIS:       TYPE = TWOLEVEL RANDOM;
                ESTIMATOR = BAYES;
                BITERATIONS = (25000);
                PROCESSORS = 2;
```

Step 5a: Two-level RDSEM Factor Analysis, Cont'd

```
MODEL:      %WITHIN%
            fpa1 BY pala1-paha1* (&1 1-4);
            fpa2 BY paha3* paha2 paha1(&1 11-13);
            fpa1-fpa2@1;
            s11 | fpa1 ON x1;
            s12 | fpa1 ON x2;
            s21 | fpa2 ON x1 ;
            s22 | fpa2 ON x2 ;
            fpa1^-fpa2^ ON fpa1^1 fpa2^1;

            %BETWEEN%
            f1b BY pala1-paha1*;
            f2b BY paha3* paha2 paha1;
            f1b-f2b@1; f1b WITH f2b (c);
            f1b f2b s11-s22 ON girl sdq age tiredavg;

OUTPUT:     STANDARDIZED TECH1 TECH4 TECH8;
```

Step 5a: Two-level RDSEM Factor Analysis, Cont'd

MODEL PRIORS:	c$\sim$ IW(0,3);
PLOT:	TYPE = PLOT3; FACTORS = ALL;
SAVEDATA:	SAVE = FSCORES(200); FACTORS = ALL; FILE = fscov imp*.dat;

Step 5b: Analysis of Amplitude and Phase Based on 200 Multiple Imputations of Step 5a

```
DATA:        FILE = fscov implist.dat;
             TYPE = IMPUTATION;

VARIABLE:    NAMES = AGE SDQ GIRL TIREDAVG
             F1B F2B S11 S12 S21 S22 B_PALA1 B_PALA2 B_PALA3
             B_PAHA1 B_PAHA2 B_PAHA3 ID;
             USEV = age-tiredavg f1b f2b amp1 amp2 phase1 phase2;
             MISSING = *;

DEFINE:      amp1 = SQRT(s11^2 + s12^2);
             amp2 = SQRT(s21^2 + s22^2);

             IF (s11>=0 .AND. s12>0) THEN phase1 = (ATAN(s11/s12))/(6.28*(1/8));
             IF (s11<0 .AND. s12>0) THEN phase1 = (6.28+ATAN(s11/s12))/(6.28*(1/8));
             IF (s11≥0 .AND. s12<0) THEN phase1 = (3.14+ATAN(s11/s12))/(6.28*(1/8));
             IF (s11<0 .AND. s12<0) THEN phase1 = (3.14+ATAN(s11/s12))/(6.28*(1/8));
             IF (s11≥0 .AND. s12==0) THEN phase1 = 3.14*0.5/(6.28*(1/8));
             IF (s11<0 .AND. s12==0) THEN phase1 = 3.14*1.5/(6.28*(1/8));
```

Step 5b: Analysis of Amplitude and Phase Based on 200 Multiple Imputations of Step 5a

```
                IF (s21≥0 .AND. s22>0) THEN phase2 = (ATAN(s21/s22))/(6.28*(1/8));
                IF (s21<0 .AND. s22>0) THEN phase2 = (6.28+ATAN(s21/s22))/(6.28*(1/8));
                IF (s21≥0 .AND. s22<0) THEN phase2 = (3.14+ATAN(s21/s22))/(6.28*(1/8));
                IF (s21<0 .AND. s22<0) THEN phase2 = (3.14+ATAN(s21/s22))/(6.28*(1/8));
                IF (s21≥0 .AND. s22==0) THEN phase2 = 3.14*0.5/(6.28*(1/8));
                IF (s21<0 .AND. s22==0) THEN phase2 = 3.14*1.5/(6.28*(1/8));
                CENTER age sdq tiredavg (GRANDMEAN);

ANALYSIS:       ESTIMATOR = BAYES;
                FBITERATIONS = 2000;

MODEL:          f1b f2b amp1 amp2 phase1 phase2 on age-tiredavg;
                [amp1] (a1);
                [amp2] (a2);
                [phase1] (p1);
                [phase2] (p2);

MODEL CONSTRAINT:
                NEW(diffAmp diffPh);
                diffAmp = a1-a2;
                diffPh = p1-p2;
```

Epilogue



- Does tiredness follow cycles due to circadian (24-hour) rhythm?
- Are such cycles also affecting PA?
- Is there a relationship between tiredness and PA after accounting for these cycles?

Analysis 2a for Tired: Cross-Classified DSEM Plot of Cycles and Between Time Estimates



- Between Time $R^2(Tired) = 0.73$ ($R^2(PA) = 0.41$)

Two-level RDSEM with two factors related to cycles and tiredness:



Input for Two-Level RDSEM with PA Factors Related to Tiredness Taking Cycles Into Account

```
          USEVARIABLES = PALA1 PALA2
          PALA3 PAHA1 PAHA2 PAHA3 tired
          x1 x2;
          CLUSTER = id;
          LAGGED = tired(1);
          WITHIN = x1 x2;
          TINTERVAL = hrs (3 time);

DEFINE:   x1 = SIN(6.2831853*(1/8)*time);
          x2 = COS(6.2831853*(1/8)*time);

ANALYSIS: TYPE = TWOLEVEL;
          ESTIMATOR = BAYES;
          BITERATIONS = (25000);
          PROCESSORS = 2;

MODEL:    %WITHIN%
          fpa1 BY pala1-paha1* (&1);
          fpa2 BY paha3* paha2 paha1 (&1);
          fpa1-fpa2@1;
          fpa1-fpa2 ON x1 x2;
          tired ON x1 x2;
          fpa1^-fpa2^ ON tired^;
          fpa1^-fpa2^ ON fpa1^1-fpa2^1;
          tired^ ON tired^1;
          %BETWEEN%
          f1b BY pala1-paha1*;
          f2b BY paha* paha2 paha1;
          f1b-f2b@1;
          f1b f2b ON tired;

OUTPUT:   STANDARDIZED TECH1 TECH4
          TECH8;
PLOT:     TYPE = PLOT3;
          FACTORS = ALL;
```

Output Excerpts for PA Factors Related to Tiredness Taking Cycles Into Account (STDYX)

	Posterior Estimate	S.D.	95% C.I. Lower 2.5%	Upper 2.5%	Significance
Within					
FPA1^ ON					
TIRED^	-0.093	0.013	-0.119	-0.067	*
FPA2^ ON					
TIRED^	-0.265	0.013	-0.289	-0.240	*
Between					
F1B ON					
TIRED	-0.450	0.063	-0.539	-0.287	*
F2B ON					
TIRED	-0.571	0.044	-0.643	-0.464	*

Monte Carlo Simulations of Cycles

- Two-level random RDSEM
- Bivariate two-level RDSEM
- Cross-classified DSEM

Monte Carlo Simulation Using Two-Level Random RDSEM Cycles Analysis with N = 200, T = 56. Step 1

```
MONTECARLO:   NAMES = pa sx1 sx2;
              NOBSERVATIONS = 11200;
              NREPS = 500;
              CSIZES = 200(56);
              NCSIZE = 1;
              LAGGED = pa(1);
              REPSAVE = ALL;
              SAVE = pa2LRandomstep1T=56Rep=500rep*.dat;
              BETWEEN = sx1 sx2;

ANALYSIS:     TYPE = TWOLEVEL;
              ESTIMATOR = BAYES;
              BITERATIONS = (200); ! complete convergence
              ! not needed
              PROCESSORS = 2;

MODEL POPULATION:   %WITHIN%
                    pa^ ON pa^1*0.37243;
                    pa*0.51090;

                    %BETWEEN%
                    pa WITH sx1*-0.00509;
                    pa WITH sx2*-0.01340;
                    sx1 WITH sx2*-0.00127;

                    [ pa*5.67306 ] ;
                    [ sx1*-0.08903 ] ;
                    [ sx2*-0.00674 ] ;

                    pa*0.74775;
                    sx1*0.01524;
                    sx2*0.00752;

MODEL:              Same as MODEL POPULATION
```

Monte Carlo Simulation Using Two-Level Random RDSEM Cycles Analysis with N = 200, T = 56. Step 2

```
DATA:        FILE = pa2LRandomstep1T=56rep=500replist.dat;
             TYPE = MONTECARLO;

VARIABLE:    NAMES = sx1 sx2 pa id time pa1;
             USEV = pa x1 x2;
             CLUSTER = id;
             LAGGED = pa(1);
             WITHIN = x1 x2;

DEFINE:      x1 = SIN(6.2831853*(1/8)*time);
             x2 = COS(6.2831853*(1/8)*time);
             pa = sx1*x1+ sx2*x2 + pa;

ANALYSIS:    TYPE = TWOLEVEL RANDOM;
             ESTIMATOR = BAYES;
             BITERATIONS = (1000);
             PROCESSORS = 2;

MODEL:       %WITHIN%

             sx1 | pa ON x1;
             sx2 | pa ON x2;

             pa^ ON pa^1*0.37243;

             pa*0.51090;

             %BETWEEN%

             pa WITH sx1*-0.00509;
             pa WITH sx2*-0.01340;
             sx1 WITH sx2*-0.00127;

             [ pa*5.67306 ] ;
             [ sx1*-0.08903 ] ;
             [ sx2*-0.00674 ] ;

             pa*0.74775;
             sx1*0.01524;
             sx2*0.00752;
```

Monte Carlo Simulation Using Bivariate Two-Level RDSEM Cycles Analysis with N = 200, T = 56. Step 1

```
MONTECARLO:     NAMES = pa tired;
                NOBSERVATIONS = 11200;
                NREPS = 500;
                CSIZES = 200(56);
                NCSIZE = 1;
                LAGGED = pa(1) tired1);
                REPSAVE = ALL;
                SAVE = pa2LBivstep1T=56Rep=500rep*.dat;

ANALYSIS:       TYPE = TWOLEVEL;
                ESTIMATOR = BAYES;
                BITERATIONS = (200); ! complete convergence
                ! not needed
                PROCESSORS = 2;

MODEL POPULATION:   %WITHIN%
                    pa^ ON pa^1*0.35281;
                    pa^ ON tired^*-0.12401;
                    tired^ ON tired^1*0.38683;

                    pa*0.49183;
                    tired*1.41098;

                    %BETWEEN%
                    pa WITH tired*-0.53406;

                    [ pa*5.66681 ];
                    [ tired*3.55581 ];

                    pa*0.74610;
                    tired*1.43987;

MODEL:              Same as MODEL POPULATION
```

Monte Carlo Simulation Using Bivariate Two-Level RDSEM Cycles Analysis with N = 200, T = 56. Step 2

```
DATA:        FILE = pa2LBivstep1T=56rep=500replist.dat;
             TYPE = MONTECARLO;

VARIABLE:    NAMES = pa tired id time pa0 pa1 tired0 tired1;
             USEV = pa tired x1 x2;
             CLUSTER = id;
             LAGGED = pa(1) tired(1);
             WITHIN = x1 x2;

DEFINE:      x1 = SIN(6.2831853*(1/8)*time);
             x2 = COS(6.2831853*(1/8)*time);
             pa = -0.09278*x1-0.01485*x2 + pa;
             tired = -0.03656*x1 + 0.52587*x2 + tired;

ANALYSIS:    TYPE = TWOLEVEL;
             ESTIMATOR = BAYES;
             BITERATIONS = (1000);
             PROCESSORS = 2;

MODEL:       %WITHIN%

             pa ON x1*-0.09278;
             pa ON x2*-0.01485;
             pa^ ON pa^1*0.35281;
             pa^ ON tired^*-0.12401;
             tired ON x1*-0.03656;
             tired ON x2*0.52587;
             tired^ ON tired^1*0.38683;

             pa*0.49183;
             tired*1.41098;

             %BETWEEN%

             pa WITH tired*-0.53406;

             [ pa*5.66681 ];
             [ tired*3.55581 ];

             pa*0.74610;
             tired*1.43987;
```

Monte Carlo Simulation Using Cross-Classified DSEM Cycles Analysis with N = 200, T = 56. Step 1

```
MONTECARLO:    NAMES = y;
               NOBSERVATIONS = 11200;
               NREPS = 500;
               CSIZES = 200[56(1)]; ! 200 subjects (2b),
               ! 56 time points (2a)
               NCSIZE = 1[1];
               LAGGED = y(1);
               REPSAVE = ALL;
               SAVE = paccstep1T=56Rep=500rep*.dat;

ANALYSIS:      TYPE = CROSSCLASSIFIED;
               ESTIMATOR = BAYES;
               BITERATIONS = (200); ! complete convergence
               ! not needed
               PROCESSORS = 2;
```

```
MODEL POPULATION:   %WITHIN%
                    y ON y&1*0.371;
                    y*0.513;

                    %BETWEEN LEVEL2A%
                    ! betweeen time
                    y*0.006;

                    %BETWEEN LEVEL2B%
                    ! between individuals
                    y*0.740; [y*5.676];
```

Monte Carlo Simulation Using Cross-Classified DSEM Cycles Analysis with N = 200, T = 56. Step 2

```
DATA:        FILE = paccstep1T=56Rep=500replist.dat;
             TYPE = MONTECARLO;

VARIABLE:    NAMES = y time id y1;
             USEV = y x1 x2;
             CLUSTER = id time;
             LAGGED = y(1);
             BETWEEN = (time) x1 x2;

DEFINE:      x1 = SIN(6.2831853*(1/8)*time);
             x2 = COS(6.2831853*(1/8)*time);
             y = -0.088*x1-0.009*x2 + y;

ANALYSIS:    TYPE = CROSSCLASSIFIED;
             ESTIMATOR = BAYES;
             BITERATIONS = (1000);
             PROCESSORS = 2;

MODEL:       %WITHIN%
             y ON y&1*0.371;
             y*0.513;

             %BETWEEN id%
             y*0.740; [y*5.676];

             %BETWEEN time%
             y*0.006;
             y ON x1*-0.088 x2*-0.009;
```